Kapitel 11

Semantik

In Kapitel 3 haben wir die programmiersprachlichen Aspekte einer Teilsprache von Standard ML betrachtet. Jetzt vertiefen wir diese Betrachtung und entwickeln eine präzise Definition für eine vereinfachte Variante dieser Teilsprache.

Die präzise Definition von Programmiersprachen ist keine ganz leichte Aufgabe. Wir sind allerdings hervorragend vorbereitet, da wir viele der benötigten Konzepte schon als programmiersprachliche Konzepte kennengelernt haben.

Man unterscheidet zwischen syntaktischen und semantischen Aspekten von Programmiersprachen. Die *Syntax* einer Programmiersprache regelt, wie man Programme darstellt. Die *Semantik* einer Programmiersprache regelt, welche Programme zulässig sind und welche Ergebnisse die Ausführung von zulässigen Programmen zu liefern hat.

Abbildung 11.1 zeigt die Zusammenhänge zwischen Syntax und Semantik. Im Zentrum steht die *abstrakte Syntax*, die die erforderlichen *syntaktischen Objekte* als mathematische Objekte definiert. Beispiele für syntaktische Objekte sind Ausdrücke, Typen und Deklarationen. Die *konkrete Syntax* regelt, wie die Objekte der abstrakten Syntax textuell, also als Folgen von Zeichen, repräsentiert werden. Die *statische Semantik* definiert Konsistenzbedingungen für die syntaktischen Objekte. Diese Bedingungen betreffen die Wohlgetyptheit und die Bindung von Bezeichnern. Die *dynamische Semantik* definiert, was bei der Ausführung von syntaktischen Objekten passieren soll.

In Kapitel 3 haben wir die drei Analysephasen eines Interpreters kennengelernt. Die lexikalische und syntaktische Analyse sind für die Übersetzung der konkreten Syntax in die abstrakte Syntax zuständig. Die sematische Analyse überprüft die Einhaltung der durch die statische Semantik auferlegten Konsistenzbedingungen.

Man unterscheidet zwischen statischen und dynamischen Aspekte von Program-



Abbildung 11.1: Syntax und Semantik

miersprachen. Die die Analysephasen betreffenden Aspekte bezeichnet man als *statisch*. Die die Ausführung betreffenden Aspekte bezeichnet man als *dynamisch*.

Idealisierte Programmiersprache F

Richtige Programmiersprachen sind sehr komplexe Artefakte. In diesem Kapitel betrachten wir daher eine kleine, idealisierte Programmiersprache, die wir F nennen. F ist eine monomorph typisierte Teilsprache von Standard ML, die trotz ihrer Einfachheit wesentliche programmiersprachliche Konzepte beinhaltet.

In diesem Kapitel entwickeln wir mathematische Definitionen für die abstrakte Syntax sowie die statische und dynamischen Semantik von F. Aus den Definitionen der statischen und dynamischen Semantik leiten wir Realisierungen für die entsprechenden Verarbeitungsphasen eines Interpreters ab. Dabei legen wir keinen Wert auf Effizienz.

11.1 Abstrakte Syntax

Abbildung 11.2 definiert die abstrakte Syntax von F mittels einer schematischen Darstellung, die man als *abstrakte Grammatik* bezeichnet. Die Grammatik definiert fünf Mengen, deren Elemente als Konstanten, Bezeichner, Operationen, Typen und Ausdrücke bezeichnet werden. Diese Mengen können auch mithilfe der Typdeklarationen in Abbildung 11.3 beschrieben werden. Mathematisch gesehen unterscheiden sich die durch die Grammatik und die Typdeklarationen definierten Mengen nur dadurch, dass die Grammatik Bezeichner als natürliche Zahlen modelliert, während die Typdeklarationen Bezeichner als Zeichenreihen realisieren. Wir erlauben uns diese kleine Abweichung, da es für die Belange der Semantik keine Rolle spielt, welche unendliche Menge wir für die Bezeichner wählen.

In Kapitel 6 haben wir die Werte von Konstruktortypen als mathematische Objekte definiert. Die Definitionen der Grammatik sind analog zu den Typdeklarationen in

c	$\in$	*Con*	$=$ `false` | `true` | $\mathbb{Z}$	*Konstanten*
x	$\in$	*Id*	$=$ $\mathbb{N}$	*Bezeichner*
o	$\in$	*Ops*	$=$ $+$ | $-$ | $*$ | $\leq$	*Operationen*
t	$\in$	*Ty*	$=$ `bool` | `int` | $t_1 \rightarrow t_2$	*Typen*
e	$\in$	*Exp*	$=$	*Ausdrücke*
			c	*Konstante*
			| x	*Bezeichner*
			| e_1 o e_2	*Operationsanwendung*
			| `if` e_1 `then` e_2 `else` e_3	*Konditional*
			| `fn` x : t `=>` e	*Abstraktion*
			| e_1 e_2	*Prozeduranwendung*

Abbildung 11.2: Abstrakte Syntax von F (Abstrakte Grammatik)

```
datatype con = False | True | IC of int

type     id  = string

datatype ops = Add | Sub | Mul | Leq

datatype ty  = Bool | Int | Arrow of ty * ty

datatype exp = Con of con
             | Id  of id
             | Op  of exp * ops * exp
             | If  of exp * exp * exp
             | Abs of id  * ty  * exp
             | App of exp * exp
```

Abbildung 11.3: Abstrakte Syntax von F (Typdeklarationen)

Abbildung 11.3 zu verstehen, wobei wir idealisierend annehmen, dass die Werte von `int` die ganzen Zahlen sind. Die Menge der Konstanten ist also wie folgt definiert:

$$Con \;=\; \{\langle 1\rangle\} \cup \{\langle 2\rangle\} \cup (\{\langle 3\rangle\} \times \mathbb{Z})$$

Die mathematische Essenz der Grammatik ist wie gesagt die Definition der Mengen *Con*, *Id*, *Ops*, *Ty* und *Exp*. Zusätzlich legt die Grammatik Notationen für die Elemente dieser Mengen fest. Diese Notationen sind unbedingt erforderlich, damit wir syntaktischen Objekte in lesbarer Form beschreiben können. Beispielsweise können wie den Ausdruck

$$\begin{aligned}&\langle 4,\ \langle\langle 3, \langle\langle 2, 1\rangle, \langle 4\rangle, \langle 1, \langle 3, 1\rangle\rangle\rangle\rangle,\\ &\quad \langle 1, \langle 3, 1\rangle\rangle,\\ &\quad \langle 3, \langle\langle 2, 1\rangle, \langle 3\rangle, \langle 6, \langle 2, 2\rangle, \langle 3, \langle\langle 2, 1\rangle, \langle 2\rangle, \langle 1, \langle 3, 1\rangle\rangle\rangle\rangle\rangle\rangle\rangle\rangle\rangle\end{aligned}$$

mit den durch die Grammatik festgelegten Notationen lesbar wie folgt beschreiben:

$$\texttt{if}\ x1 \leq 1\ \texttt{then}\ 1\ \texttt{else}\ x1 * x2\,(x1 - 1)$$

Die Typdeklaration in Abbildung 11.3 legen alternative Notationen für die syntaktische Objekte fest. Damit kann der obige Ausdruck wie folgt beschrieben werden:

```
If(Op(Id "x1", Leq, Con(IC 1)),
   Con(IC 1),
   Op(Id "x1", Mul, App(Id "x2", Op(Id "x1", Sub, Con(IC 1)))))
```

Die durch die Grammatik festgelegten mathematischen Notationen sind sehr kompakt. Das wird mit diversen notationalen Tricks erkauft, die auf die Intelligenz des menschlichen Lesers vertrauen. Zum Beispiel wird auf einige der eigentlich notwendige Konstruktoren verzichtet (die Typdeklarationen führen alle Konstruktoren ein). Die fehlenden Konstruktoren muss der menschliche Leser aus dem Kontext ableiten und an den richtigen Stellen einfügen.

Die Grammatik legt nicht fest, ob die Beschreibung

$$1 + 2 + 3$$

als der Ausdruck $(1 + 2) + 3$ oder der Ausdruck $1 + (2 + 3)$ zu interpretieren ist. Dieses Problem kann man dadurch lösen, dass man zusätzliche Interpretationsregeln angibt. Wir verzichten hier auf solche Regeln und sorgen stattdessen mithilfe von Klammern dafür, dass die verwendeten Beschreibungen eindeutig interpretiert werden können.




```
fn x:int => fn y:int->int => z (x+3) (y x)
```

Abbildung 11.4: Baumdarstellung eines Ausdrucks

avariablen

Die in der Grammatik verwendeten Buchstaben *c*, *x*, *o*, *t* und *e* dienen als notationale Variablen und werden als *Metavariablen* bezeichnet. Im Kontext der Grammatik bezeichnet eine Metavariable stets ein beliebiges Element der ihr durch die Grammatik zugeordneten Menge. Beispielsweise bezeichnet *t* im Kontext der Grammatik stets einen Typ.

Wir haben jetzt zum ersten Mal präzise gesagt, um welche mathematischen Objekte es sich bei Ausdrücken und Typen handelt. Die dazu notwendigen Konzepte haben wir bereits im Zusammenhang mit Konstruktortypen kennengelernt.

Die Formalisierung von Ausdrücken als Tupel macht die Baumstruktur von Ausdrücken explizit. Es ist oft nützlich, die Baumstruktur eines Ausdrucks grafisch darzustellen. Abbildung 11.4 zeigt ein Beispiel.

Die durch die Grammatik festgelegten Notationen suggerieren, wie wir F als Teilsprache von Standard ML auffassen können. Wir werden die statische und dynamische Semantik von F genau gemäß dieser Interpretation definieren. Das hat den Vorteil, dass wir F nicht weiter erklären müssen und eine klare Vorstellung davon haben, was wir definieren wollen. Außerdem können wir für Beispiele die konkrete Syntax von Standard ML für F benutzen.

Wir werden F ohne Größenbeschränkungen für Zahlen und ohne Speicherbeschränkungen definieren. Auch die Definition von Standard ML verzichtet aus guten Gründen auf diese Beschränkungen. Es ist durchaus möglich, Standard ML ohne Größenbeschränkungen für ganze Zahlen zu implementieren (siehe Abschnitt 10.2). Dagegen sind Speicherbeschränkungen von prinzipieller Natur. Sie sind aber abhängig von dem jeweiligen Computer, auf dem ein Programm ausgeführt wird, und der Art und Weise, wie ein Programmiersystem den zur Verfügung stehenden Speicher verwaltet.

11.2 Statische Semantik

Die statische Semantik definiert Konsistenzbedingungen für die syntaktischen Objekte. Diese Bedingungen betreffen die Wohlgetyptheit und die Bindung von Bezeichnern.

Im Kontext der statischen Semantik werden Bezeichner an Typen gebunden. Eine Menge solcher Bindungen bezeichnet man als eine Umgebung. Genauer verstehen wir unter einer *Typumgebung* (englisch „type environment“) eine endliche Funktion, die Bezeichner auf Typen abbildet. Wir verwenden die folgenden Schreibweisen:

$$T \in TE = Id \xrightarrow{fin} Ty$$

Wir definieren die statische Semantik von F als eine Menge

$$SS \subseteq TE \times Exp \times Ty$$

und schreiben

$$T \vdash e \Rightarrow t$$

für

$$(T,\ e,\ t) \in SS$$

Dabei soll $T \vdash e \Rightarrow t$ genau dann gelten, wenn der Ausdruck e in der Typumgebung T wohlgetypt ist und den Typ t hat. Die Typumgebung T ist dafür zuständig, den in e frei auftretenden Bezeichnern Typen zuzuordnen.

Die Menge SS definieren wir rekursiv durch die Inferenzregeln in Abbildung 11.5. Die Notation $T + \{x \mapsto t\}$ wurde in Abschnitt 2.3 definiert. Für jede der sechs Ausdrucksvarianten gibt es mindestens eine Regel. Die in diesen Regeln vorkommenden Metavariablen (zum Beispiel T oder c) dürfen dabei nur an Objekte der jeweils festgelegten Klasse gebunden werden. Die Definition der statischen Semantik durch die Inferenzregeln ist kompakt und modular (jede Ausdrucksvariante wird für sich behandelt). Machen Sie sich klar, dass die Regeln neben der Typanalyse auch die notwendige Bindungsanalyse formulieren.

Die folgende Proposition stellt fest, dass ein Ausdruck in einer Typumgebung höchstens einen Typ hat.

Proposition 11.2.1 (Determinismus) *Sei $T \vdash e \Rightarrow t$ und $T \vdash e \Rightarrow t'$. Dann gilt $t = t'$.*

$$\frac{}{T \vdash \texttt{false} \Rightarrow \texttt{bool}} \qquad \frac{}{T \vdash \texttt{true} \Rightarrow \texttt{bool}}$$

$$\frac{c \in \mathbb{Z}}{T \vdash c \Rightarrow \texttt{int}} \qquad \frac{T(x) = t}{T \vdash x \Rightarrow t}$$

$$\frac{o \in \{+, -, *\} \quad T \vdash e_1 \Rightarrow \texttt{int} \quad T \vdash e_2 \Rightarrow \texttt{int}}{T \vdash e_1\ o\ e_2 \Rightarrow \texttt{int}}$$

$$\frac{T \vdash e_1 \Rightarrow \texttt{int} \quad T \vdash e_2 \Rightarrow \texttt{int}}{T \vdash e_1 \leq e_2 \Rightarrow \texttt{bool}}$$

$$\frac{T \vdash e_1 \Rightarrow \texttt{bool} \quad T \vdash e_2 \Rightarrow t \quad T \vdash e_3 \Rightarrow t}{T \vdash \texttt{if}\ e_1\ \texttt{then}\ e_2\ \texttt{else}\ e_3 \Rightarrow t}$$

$$\frac{T + \{x \mapsto t\} \vdash e \Rightarrow t'}{T \vdash \texttt{fn}\ x : t\ \texttt{=>}\ e \Rightarrow t \to t'}$$

$$\frac{T \vdash e_1 \Rightarrow t' \to t \quad T \vdash e_2 \Rightarrow t'}{T \vdash e_1\ e_2 \Rightarrow t}$$

Abbildung 11.5: Statische Semantik von F

```
type 'a env
exception Unbound of id
val empty  : 'a env
val insert : id * 'a * 'a env -> 'a env
val lookup : id * 'a env -> 'a  (* Unbound *)
```

Abbildung 11.6: Signatur für Umgebungen

Ein Ausdruck e heißt *zulässig* genau dann, wenn es einen Typ t gibt mit $\emptyset \vdash e \Rightarrow t$.

Proposition 11.2.2 *Jeder zulässige Ausdruck ist geschlossen.*

Man kann die Regeln der statischen Semantik direkt in eine rekursive Prozedur `elab` umformulieren, die zu einer Typumgebung T und zu einem Ausdruck e feststellt, ob ein Typ t existiert, so dass $T \vdash e \Rightarrow t$ gilt. Wenn ein Typ existiert, wird er als Ergebnis geliefert. Wenn kein Typ existiert, wird eine Ausnahme geworfen, die beschreibt, welcher statische Fehler vorliegt.

Wir nehmen an, dass Umgebungen gemäß der Signatur in Abbildung 11.6 realisiert sind. Abbildung 11.7 zeigt die Realisierung der Prozedur `elab`. Diese ist mit einer Fallunterscheidung nach den sechs Ausdrucksvarianten realisiert. Jeder Fall realisiert die entsprechenden Inferenzregeln. Die rekursiven Aufrufe werden durch die Prämissen der Regeln beschrieben. Die Rekursion terminiert, da die Ausdrücke in den Prämissen einer Regel stets kleiner sind als der Ausdruck in der Konklusion der Regel.

11.3 Dynamische Semantik

Die dynamische Semantik legt fest, ob und mit welchem Wert die Auswertung eines Ausdrucks terminiert.

F hat zwei Arten von Werten, ganze Zahlen und Prozeduren. Die Booleschen Werte stellen wir, wie in Abschnitt 3.1 besprochen, durch die Zahlen 0 und 1 dar.

In Abschnitt 3.9 haben wir gelernt, dass man Prozeduren durch geschlossene Abstraktionen darstellen kann. Hier wählen wir eine alternative Darstellung, die mit sogenannten *Wertumgebungen* arbeitet. Wertumgebungen binden Bezeichner an Werte und spielen in der dynamischen Semantik eine ähnliche Rolle wie Typumgebungen in der statischen Semantik.

```
exception Error of string

fun elabCon True   = Bool
  | elabCon False  = Bool
  | elabCon (IC _) = Int

fun elabOp Int Add Int = Int
  | elabOp Int Sub Int = Int
  | elabOp Int Mul Int = Int
  | elabOp Int Leq Int = Bool
  | elabOp _   _   _   = raise Error "ill-typed operation"

fun elab T (Con c)         = elabCon c

  | elab T (Id  s)         = lookup(s,T)

  | elab T (Op(e1,ops,e2)) = elabOp (elab T e1) ops (elab T e2)

  | elab T (If(e1,e2,e3))  = let val t1 = elab T e1
                                 val t2 = elab T e2
                                 val t3 = elab T e3
                             in  if t1=Bool andalso t2=t3 then t2
                                 else raise
                                    Error "ill-typed conditional"
                             end

  | elab T (App(e1,e2))    = let val t1 = elab T e1
                                 val t2 = elab T e2
                                 val s  = "ill-typed application"
                             in  case t1 of
                                Arrow(t,t') => if t=t2 then t'
                                               else raise Error s
                                | _  => raise Error s
                             end

  | elab T (Abs(s,t,e))    = Arrow(t, elab (insert(s,t,T)) e)

val elab  : ty env -> exp -> ty  (* Error, Unbound *)
```

Abbildung 11.7: Die Prozedur `elab`

$$
\begin{array}{lcll}
v \in Val & = & \mathbb{Z} \cup Pro & \textit{Werte} \\
Pro & = & Id \times Exp \times VE & \textit{Prozeduren} \\
V \in VE & = & Id \xrightarrow{fin} Val & \textit{Wertumgebungen}
\end{array}
$$

Abbildung 11.8: Werte, Prozeduren und Wertumgebungen für F

Prozeduren sind Werte, die durch die Auswertung von Abstraktionen ins Spiel kommen. Wir stellen die Prozedur, die durch die Auswertung einer Abstraktion

`fn` x : t `=>` e

erhalten wird, durch ein Tripel

$$(x,\ e,\ V)$$

dar, dessen dritte Komponente V eine Wertumgebung ist, die den in der Abstraktion frei auftretenden Bezeichnern Werte zuordnet. Mit dieser Darstellung liefert die Auswertung des Ausdrucks

```
(fn x:int => fn y:int => x+y) 7
```

die Prozedur

$$(y,\ x + y,\ \{x \mapsto 7\})$$

Beachten Sie, dass die Typen der Argumentvariablen nicht in die Darstellung der Prozedur aufgenommen werden. Das liegt daran, dass sie bei der Anwendung der Prozedur nicht benötigt werden.

Die *Werte* (englisch „values“), *Prozeduren* (englisch „procedures“) und *Wertumgebungen* (englisch „value environments“) für die dynamische Semantik von F definieren wir rekursiv wie in Abbildung 11.8 gezeigt.

Wir definieren die dynamische Semantik von F als eine Menge

$$DS \subseteq VE \times Exp \times Val$$

und schreiben

$$V \vdash e \Rightarrow v$$

für

$$(V,\ e,\ v) \in DS$$

Dabei soll $V \vdash e \Rightarrow v$ genau dann gelten, wenn die Evaluation des Ausdrucks e in der Wertumgebung V terminiert und den Wert v liefert. Die Wertumgebung V ist dafür zuständig, den in e frei auftretenden Bezeichnern Werte zuzuordnen.

Die Menge *DS* definieren wir rekursiv durch die Inferenzregeln in Abbildung 11.9. Die Regel für Abstraktionen liefert Prozeduren, die die gesamte Umgebung mit sich tragen, die bei der Auswertung der Abstraktion vorlag. Das ist die einfachste Lösung. Alternativ könnten wir einer Prozedur nur den Teil der Umgebung mitgeben, der die in der Abstraktion frei auftretenden Bezeichner bindet.

Die folgende Proposition sagt, dass das Ergebnis der Auswertung eines Ausdrucks eindeutig bestimmt ist:

Proposition 11.3.1 (Determinismus) *Sei $V \vdash e \Rightarrow v$ und $V \vdash e \Rightarrow v'$. Dann gilt $v = v'$.*

Der folgende Satz stellt fest, dass es in F weder Divergenz noch dynamische Fehler gibt.

Satz 11.3.2 (Auswertbarkeit) *Für jeden zulässigen Ausdruck e existiert genau ein Wert v mit $\emptyset \vdash e \Rightarrow v$.*

Der Auswertbarkeitssatz ist keineswegs einfach zu beweisen. Er bestätigt unsere Intuition, dass man ohne Rekursion keinen zulässigen Ausdruck angeben kann, dessen Auswertung divergiert.

Satz 11.3.3 (Typerhaltung)

1. *Sei $\emptyset \vdash e \Rightarrow$ `int` und $\emptyset \vdash e \Rightarrow v$. Dann $v \in \mathbb{Z}$.*
2. *Sei $\emptyset \vdash e \Rightarrow$ `bool` und $\emptyset \vdash e \Rightarrow v$. Dann $v \in \{0, 1\}$.*

Zulässige Ausdrücke der Typen `int` und `bool` evaluieren also immer zu Werten dieser Typen.

Der Auswertbarkeitssatz und der Typerhaltungssatz liefern wichtige Sicherheitsgarantien für zulässige Ausdrücke.

Die Werte eines Typs t können wir wie folgt definieren:

$$\mathit{Value}(t) = \{v \in \mathit{Val} \mid \exists e\colon\ \emptyset \vdash e \Rightarrow t \ \wedge\ \emptyset \vdash e \Rightarrow v\}$$

Offensichtlich gibt es in *Pro* „unechte“ Prozeduren, die zu keinem Typ gehören, zum Beispiel $(x,\ y\ x,\ \{y \mapsto 1\})$.

$$\frac{}{V \vdash \texttt{false} \Rightarrow 0} \qquad \frac{}{V \vdash \texttt{true} \Rightarrow 1} \qquad \frac{c \in \mathbb{Z}}{V \vdash c \Rightarrow c}$$

$$\frac{V(x) = v}{V \vdash x \Rightarrow v}$$

$$\frac{V \vdash e_1 \Rightarrow v_1 \quad V \vdash e_2 \Rightarrow v_2 \quad v_1, v_2 \in \mathbb{Z} \quad v = v_1 + v_2}{V \vdash e_1 + e_2 \Rightarrow v}$$

$$\frac{V \vdash e_1 \Rightarrow v_1 \quad V \vdash e_2 \Rightarrow v_2 \quad v_1, v_2 \in \mathbb{Z} \quad v = v_1 - v_2}{V \vdash e_1 - e_2 \Rightarrow v}$$

$$\frac{V \vdash e_1 \Rightarrow v_1 \quad V \vdash e_2 \Rightarrow v_2 \quad v_1, v_2 \in \mathbb{Z} \quad v = v_1 \cdot v_2}{V \vdash e_1 * e_2 \Rightarrow v}$$

$$\frac{V \vdash e_1 \Rightarrow v_1 \quad V \vdash e_2 \Rightarrow v_2 \quad v_1, v_2 \in \mathbb{Z} \quad v = \text{if } v_1 \leq v_2 \text{ then 1 else 0}}{V \vdash e_1 \leq e_2 \Rightarrow v}$$

$$\frac{V \vdash e_1 \Rightarrow 1 \qquad V \vdash e_2 \Rightarrow v}{V \vdash \texttt{if}\ e_1\ \texttt{then}\ e_2\ \texttt{else}\ e_3 \Rightarrow v}$$

$$\frac{V \vdash e_1 \Rightarrow 0 \qquad V \vdash e_3 \Rightarrow v}{V \vdash \texttt{if}\ e_1\ \texttt{then}\ e_2\ \texttt{else}\ e_3 \Rightarrow v}$$

$$\frac{}{V \vdash \texttt{fn}\ x : t\ \texttt{=>}\ e \Rightarrow (x, e, V)}$$

$$\frac{V \vdash e_1 \Rightarrow (x, e, V') \quad V \vdash e_2 \Rightarrow v_2 \quad V' + \{x \mapsto v_2\} \vdash e \Rightarrow v}{V \vdash e_1\ e_2 \Rightarrow v}$$

Abbildung 11.9: Dynamische Semantik von F

Beachten Sie, dass die Inferenzregeln in Abbildung 11.9 die Ordnung, in der Teilausdrücke ausgewertet werden sollen, nur teilweise festlegen. Beispielsweise lässt die Regel für Addition

$$\frac{V \vdash e_1 \Rightarrow v_1 \quad V \vdash e_2 \Rightarrow v_2 \quad v_1, v_2 \in \mathbb{Z} \quad v = v_1 + v_2}{V \vdash e_1 + e_2 \Rightarrow v}$$

offen, ob e_1 oder e_2 zuerst ausgewertet werden soll.

Die Regeln der dynamischen Semantik kann man wie bei der statischen Semantik zu einer rekursiven Prozedur

```
eval : value env -> exp -> value
```

umformulieren, die einen Ausdruck e in einer Wertumgebung V evaluiert und seinen Wert als Ergebnis liefert. Abbildung 11.10 zeigt die Realisierung der Prozedur `eval`. Der Auswertbarkeitssatz 11.3.2 garantiert, dass `eval` für zulässige Ausdrücke immer regulär terminiert (also ohne eine Ausnahme zu werfen). Der Typerhaltungssatz 11.3.3 garantiert, dass `eval` für zulässige Ausdrücke des Typs `int` [`bool`] immer ein Ergebnis der Form `IV` *n* [`IV 0` oder `IV 1`] liefert.

Die dynamische Semantik ist unabhängig von der statischen Semantik definiert. Die definierenden Inferenzregeln und die Prozedur `eval` ignorieren die in den Abstraktionen vermerkten Typen für die Argumentvariablen völlig. Für die dynamische Semantik spielt es also keine Rolle, welche Typen man für die Argumentvariablen angibt. Man hat also die Möglichkeit, die in einem Ausdruck angegebenen Typen vor der Ausführung vollständig zu löschen. Von dieser Möglichkeit machen Programmiersysteme für SML aus Effizienzgründen in der Tat Gebrauch.

11.4 Einschritt-Semantik

In den Abschnitten 1.3 und 3.9 haben wir die Auswertung von Ausdrücken mithilfe von Auswertungsprotokollen beschrieben. Wir definieren jetzt die diesen Protokollen zugrundeliegenden Auswertungsschritte präzise. Eine solche Definition bezeichnet man als *Einschritt-Semantik*.

Ein geschlossener Ausdruck heißt *kanonisch* genau dann, wenn er eine Konstante oder eine Abstraktion ist. Kanonische Ausdrücke sind bei der Einschrittsemantik nicht weiter auswertbar und dienen als Darstellungen für Werte.

Die Einschritt-Semantik definiert eine Relation

$$\rightarrow \subseteq \mathit{Exp} \times \mathit{Exp}$$

```
datatype value = IV    of int
               | Proc  of id * exp * value env

fun evalCon False  = IV 0
  | evalCon True   = IV 1
  | evalCon (IC x) = IV x

fun evalOp (IV x1) Add (IV x2) = IV(x1+x2)
  | evalOp (IV x1) Sub (IV x2) = IV(x1-x2)
  | evalOp (IV x1) Mul (IV x2) = IV(x1*x2)
  | evalOp (IV x1) Leq (IV x2) = IV(if x1<=x2 then 1 else 0)
  | evalOp  _      _   _       = raise Error "type error"

fun eval V (Con c)         = evalCon c

  | eval V (Id  s)         = lookup(s,V)

  | eval V (Op(e1,ops,e2)) = evalOp (eval V e1) ops (eval V e2)

  | eval V (If(e1,e2,e3))  = (case eval V e1 of
                                IV 1 => eval V e2
                              | IV 0 => eval V e3
                              |  _   => raise Error "type error")

  | eval V (Abs(s,_,e))    = Proc(s,e,V)

  | eval V (App(e1,e2))    = (case (eval V e1, eval V e2) of
                                (Proc(s,e,V'), v) =>
                                  eval (insert(s,v,V')) e
                              | _ => raise Error "type error")

val eval  : value env -> exp -> value  (* Error, Unbound *)
```

Abbildung 11.10: Die Prozedur `eval`

$$
\begin{aligned}
c\,[e/x] &= c \\
x'\,[e/x] &= \text{if } x = x' \text{ then } e \text{ else } x' \\
(e_1\ o\ e_2)\,[e/x] &= (e_1\,[e/x])\ o\ (e_2\,[e/x]) \\
(\texttt{if } e_1 \texttt{ then } e_2 \texttt{ else } e_3)\,[e/x] &= \texttt{if } e_1\,[e/x] \texttt{ then } e_2\,[e/x] \texttt{ else } e_3\,[e/x] \\
(\texttt{fn } x' \texttt{: } t \texttt{ => } e')\,[e/x] &= \text{if } x = x' \text{ then } \texttt{fn } x' \texttt{: } t \texttt{ => } e' \\
&\qquad\qquad \text{else } \texttt{fn } x' \texttt{: } t \texttt{ => } (e'\,[e/x]) \\
(e_1\ e_2)\,[e/x] &= (e_1\,[e/x])\ (e_2\,[e/x])
\end{aligned}
$$

Abbildung 11.11: Substitution für F

Für $(e, e') \in \rightarrow$ schreiben wir kürzer $e \rightarrow e'$. Dabei soll $e \rightarrow e'$ genau dann gelten, wenn e durch einen Auswertungsschritt in e' überführt werden kann.

Um die Auswertung von Prozeduranwendungen

$$(\texttt{fn } x\texttt{: } t \texttt{ => } e)\ e'$$

zu formalisieren, benötigen wir eine Operation, die alle freien Auftreten des Bezeichners x im Rumpf e der Abstraktion durch den kanonischen Ausdruck e' ersetzt. Eine solche Operation bezeichnet man als *Substitutionsoperation*. Den aus e durch die Substitution erhaltenen Ausdruck bezeichnet man mit $e\,[e'/x]$ (sprich „e mit e' für x“). Die Substitutionsoperation ist eine Funktion

$$Exp \times Exp \times Id \rightarrow Exp$$

Abbildung 11.11 zeigt eine rekursive Definition der Substitutionsoperation.[1]

Abbildung 11.12 zeigt eine rekursive Definition der Relation $\rightarrow$ mithilfe von Inferenzregeln. Die ersten acht Inferenzregeln heißen *Auswertungsregeln*. Die restlichen fünf Inferenzregeln heißen *Abstiegsregeln* und legen fest, welche Teilausdrücke eines Ausdrucks ausgewertet werden dürfen.

Im Gegensatz zur Ergebnissemantik legt die Einschrittsemantik die Ordnung, in der Teilausdrücke ausgewertet werden, vollständig fest. Das ist eine Voraussetzung für die Gültigkeit der folgenden Proposition:

Proposition 11.4.1 (Determinismus) *Sei $e \rightarrow e'$ und $e \rightarrow e''$. Dann $e' = e''$.*

Proposition 11.4.2 *Sei e geschlossen und $e \rightarrow e'$. Dann ist e' geschlossen.*

[1] Für Experten: Unsere Substitutionsfunktion ist naiv, da sie Kapern zulässt. Das führt zu keinen Problemen, da uns die Einschrittsemantik nur für geschlossene Ausdrücke interessiert.

$$\frac{c_1 \in \mathbb{Z} \quad c_2 \in \mathbb{Z} \quad c = c_1 + c_2}{c_1 + c_2 \;\to\; c} \qquad \frac{c_1 \in \mathbb{Z} \quad c_2 \in \mathbb{Z} \quad c = c_1 - c_2}{c_1 - c_2 \;\to\; c}$$

$$\frac{c_1 \in \mathbb{Z} \quad c_2 \in \mathbb{Z} \quad c = c_1 \cdot c_2}{c_1 * c_2 \;\to\; c}$$

$$\frac{c_1 \in \mathbb{Z} \quad c_2 \in \mathbb{Z} \quad c_1 \leq c_2}{c_1 \leq c_2 \;\to\; \texttt{true}} \qquad \frac{c_1 \in \mathbb{Z} \quad c_2 \in \mathbb{Z} \quad c_1 > c_2}{c_1 \leq c_2 \;\to\; \texttt{false}}$$

$$\frac{}{\texttt{if true then } e_1 \texttt{ else } e_2 \;\to\; e_1}$$

$$\frac{}{\texttt{if false then } e_1 \texttt{ else } e_2 \;\to\; e_2}$$

$$\frac{e' \text{ kanonisch} \quad e'' = e\,[e'/x]}{(\texttt{fn } x : t \texttt{ => } e)\; e' \;\to\; e''}$$

$$\frac{e_1 \;\to\; e_1'}{e_1 \; o \; e_2 \;\to\; e_1' \; o \; e_2} \qquad \frac{e_1 \text{ kanonisch} \quad e_2 \;\to\; e_2'}{e_1 \; o \; e_2 \;\to\; e_1 \; o \; e_2'}$$

$$\frac{e_1 \;\to\; e_1'}{\texttt{if } e_1 \texttt{ then } e_2 \texttt{ else } e_3 \;\to\; \texttt{if } e_1' \texttt{ then } e_2 \texttt{ else } e_3}$$

$$\frac{e_1 \;\to\; e_1'}{e_1\; e_2 \;\to\; e_1'\; e_2} \qquad \frac{e_1 \text{ kanonisch} \quad e_2 \;\to\; e_2'}{e_1\; e_2 \;\to\; e_1\; e_2'}$$

Abbildung 11.12: Einschrittsemantik für F

Ein Ausdruck e heißt *reduzierbar* genau dann, wenn es einen Ausdruck e' gibt mit $e \to e'$.

Proposition 11.4.3 (Reduzierbarkeit) *Sei e ein zulässiger Ausdruck. Dann ist e kanonisch genau dann, wenn e nicht reduzierbar ist.*

Für jeden zulässigen Ausdruck, der nicht kanonisch ist, gibt es also genau eine Möglichkeit, ihn mit einer Auswertungsregel zu reduzieren.

Satz 11.4.4 (Terminierung) *Sei e ein zulässiger Ausdruck. Dann gibt es keine unendliche Kette* $e \to e_1 \to e_2 \to \cdots$.

Wir schreiben $e \to^* e'$ genau dann, wenn es Ausdrücke $e_1, \ldots, e_n$ gibt mit $e = e_1 \to \cdots \to e_n = e'$. Die Relation $\to^*$ wird als die *reflexive und transitive Hülle* der Relation $\to$ bezeichnet.

Proposition 11.4.5 (Typerhaltung) *Sei* $\emptyset \vdash e \Rightarrow t$ *und* $e \to^* e'$. *Dann gilt* $\emptyset \vdash e' \Rightarrow t$.

Ein kanonischer Ausdruck e' heißt *Normalform* eines Ausdrucks e genau dann, wenn $e \to^* e'$ gilt.

Satz 11.4.6 (Normalisierbarkeit) *Jeder zulässige Ausdruck hat genau eine Normalform.*

Beweis Folgt sofort aus Satz 11.4.4 und den Propositionen 11.4.5, 11.4.3 und 11.4.1. □

Die in Abschnitt 11.3 dargestellte Definition der dynamischen Semantik von F bezeichnet man als *Ergebnissemantik,* da sie Ausdrücken direkt die Ergebisse ihrer Auswertung zuordnet. Der folgende Satz bestätigt unsere Intuition, dass die Ergebnis- und die Einschrittsemantik im Wesentlichen übereinstimmen.

Satz 11.4.7 (Übereinstimmung) *Sei* $\emptyset \vdash e \Rightarrow$ `int` *und* $n \in \mathbb{Z}$. *Dann gilt* $\emptyset \vdash e \Rightarrow n$ *genau dann, wenn* $e \to^* n$.

Man kann also die Ergebnissemantik als eine alternative Möglichkeit für die Definition der dynamischen Semantik ansehen.

Genau wie die Ergebnissemantik macht die Einschrittsemantik keinen Gebrauch von den für die Argumentvariablen angegebenen Typen. Man kann also auch

unzulässige Ausdrücke auswerten. Betrachten Sie den geschlossenen aber unzulässigen Ausdruck

$$\mathit{self} \;=\; \texttt{fn}\; x : \texttt{int} \;\texttt{=>}\; x\; x$$

Wegen der *Selbstapplikation* $x\ x$ bleibt der Ausdruck *self* auch dann unzulässig, wenn man für x andere Typen als `int` wählt. Folglich weist Moscow ML den Ausdruck

```
fn x => x x
```

als nicht typisierbar zurück. Mit *self* kann man einen divergierenden Ausdruck angeben:

$$\mathit{self}\ \mathit{self} \;\to\; \mathit{self}\ \mathit{self} \;\to\; \mathit{self}\ \mathit{self} \;\to\; \cdots$$

Das zeigt, dass die durch den Terminierungssatz und den Normalisierbarkeitssatz formulierten Sicherheitsgarantien nur für zulässige Ausdrücke gelten. Mithilfe von Selbstapplikation kann man übrigens rekursive Prozeduren simulieren (siehe Aufgabe 11.10).

11.5 Let-Ausdrücke und homogene Paare

Wir zeigen jetzt, wie man in F Let-Ausdrücke und Paare ausdrücken kann.

Zunächst stellen wir fest, dass man einen Let-Ausdruck

$$\texttt{let val}\; x : t = e_1 \;\texttt{in}\; e_2 \;\texttt{end}$$

auch mit einer Abstraktion und einer Applikation ausdrücken kann:

$$(\texttt{fn}\; x : t \;\texttt{=>}\; e_2)\; e_1$$

Let-Ausdrücke können also zu F als abgeleitete Form (siehe Abschnitt 6.4) hinzugefügt werden. Das gibt uns die Möglichkeit, die konkrete Syntax von F mit Let-Ausdrücken auszustatten, ohne die abstrakte Syntax zu erweitern. Die syntaktische Analyse übernimmt dann die Übersetzung von Let-Ausdrücken in die entsprechenden Abstraktionen und Applikationen.

Auch Let-Ausdrücke mit mehr als einer Deklaration können als abgeleitete Form zu F hinzugefügt werden. Beispielsweise können wir einen Let-Asudruck

$$\texttt{let}\; d_1\; d_2 \;\texttt{in}\; e \;\texttt{end}$$

mit zwei Deklarationen auf zwei ineinander geschachtelte Let-Ausdrücke mit je einer Deklaration zurückführen:

$$\texttt{let}\ d_1\ \texttt{in let}\ d_2\ \texttt{in}\ e\ \texttt{end end}$$

Das Hinzufügen von Let-Ausdrücken als abgeleitete Form hat den Nachteil, dass der Programmierer den Typ der deklarierten Variable explizit angeben muss. Wenn wir Let-Ausdrücke in die abstrakte Syntax aufnehmen, können wir wegen Proposition 11.2.1 auf die Angabe eines Typs verzichten. Hier sind die dann erforderlichen Regeln für die statische und dynamische Semantik:

$$\frac{T \vdash e_1 \Rightarrow t_1 \qquad T + \{x \mapsto t_1\} \vdash e_2 \Rightarrow t}{T \vdash \texttt{let val}\ x = e_1\ \texttt{in}\ e_2\ \texttt{end} \Rightarrow t}$$

$$\frac{V \vdash e_1 \Rightarrow v_1 \qquad V + \{x \mapsto v_1\} \vdash e_2 \Rightarrow v}{V \vdash \texttt{let val}\ x = e_1\ \texttt{in}\ e_2\ \texttt{end} \Rightarrow v}$$

Offensichtlich können wir in F Paare des Typs `int*int` durch Prozeduren des Typs `bool` $\rightarrow$ `int` darstellen. Diese liefern für `true` die erste und für `false` die zweite Komponente des dargestellten Paars. Hier sind Deklarationen für Prozeduren, die Paare konstruieren und zerlegen:

```
val pair = fn x:int => fn y:int => fn b:bool =>
                                      if b then x else y

val fst = fn p:bool->int => p true

val snd = fn p:bool->int => p false
```

Mit der gerade gezeigten Methode können wir homogene Paare des Typs $t * t$ darstellen. Heterogene Paare des Typs $t_1 * t_2$ mit $t_1 \neq t_2$ lassen sich dagegen nicht in F ausdrücken, da die für eine Darstellung erforderlichen Typen fehlen.

11.6 Rekursive Prozeduren

Wir erweitern F jetzt um rekursive Prozeduren. Die erweiterte Sprache nennen wir FR. Für rekursive Prozeduren benötigen wir zusätzliche Syntax. Diese wählen wir anders als in Standard ML. Wir erweitern die abstrakte Syntax von F um die neue Ausdrucksvariante

$$e \in \mathit{Exp} \ = \ \cdots \ | \ \texttt{rec}\ x_1(x_2 : t_2) : t_1\ \texttt{=>}\ e$$

die wir *rekursive Abstraktion* nennen. Entsprechend erweitern wir die Deklaration von `exp` um

```
datatype exp = ··· | Rec of id * id * ty * ty * exp
```

Die Bedeutung der neuen Ausdrucksvariante können wir in Standard ML wie folgt beschreiben:

```
let fun x1 (x2:t2): t1 = e in x1 end
```

Eine rekursive Prozedur, die die Fakultätsfunktion berechnet, können wir mit einer rekursiven Abstraktion wie folgt schreiben:

```
rec f(n:int):int => if n<=1 then 1 else n*(f(n-1))
```

Wenn wir Standard ML um rekursive Abstraktionen erweitern würden, könnten wir Prozedurdeklarationen der Form

```
fun x1 (x2:t2): t1 = e
```

auf Wertdeklarationen der Form

```
val x1 = rec x1 (x2:t2): t1 => e
```

zurückführen.

Die rekursive Abstraktion verlangt einen Ergebnistyp, da dieser im Allgemeinen durch den Argumenttyp nicht eindeutig festgelegt ist. Hier ist ein Beispiel:

```
rec f (x:int) : int => f x
```

Ein alternativer, ebenfalls zulässiger Ergebnistyp wäre `bool`:

```
rec f (x:int) : bool => f x
```

Die statische Semantik erweitern wir um die Inferenzregel:

$$\frac{T + \{x_1 \mapsto t_2 \to t_1\} + \{x_2 \mapsto t_2\} \vdash e \Rightarrow t_1}{T \vdash \texttt{rec}\ x_1(x_2 : t_2) : t_1\ \texttt{=>}\ e \Rightarrow t_2 \to t_1}$$

Die Prozedur `elab` wird entsprechend erweitert.

Um die dynamische Semantik um rekursive Prozeduren zu erweitern, müssen wir zuerst rekursive Prozeduren als Werte darstellen. Dazu wählen wir vierstellige Tupel:

$$Val = \mathbb{Z} \cup Pro \cup RPro$$

$$RPro = Id \times Id \times Exp \times VE$$

Die Bedeutung dieser Darstellung wird durch die Inferenzregel für rekursive Abstraktionen klar:

$$\frac{}{V \vdash \texttt{rec}\ x_1(x_2 : t_2) : t_1\ \texttt{=>}\ e \Rightarrow (x_1, x_2, e, V)}$$

Zusätzlich benötigen wir eine Inferenzregel für die Anwendung von rekursiven Prozeduren:

$$\frac{V \vdash e_1 \Rightarrow v_1 \quad v_1 = (x_1, x_2, e, V') \quad V \vdash e_2 \Rightarrow v_2 \quad V' + \{x_1 \mapsto v_1\} + \{x_2 \mapsto v_2\} \vdash e \Rightarrow v}{V \vdash e_1\ e_2 \Rightarrow v}$$

Die Erweiterung der Prozedur `eval` erfolgt entsprechend. Dazu erweitern wir die Deklaration von `value` um

```
datatype value = ··· | RProc of id * id * exp * value env
```

Für die Erweiterung der Einschrittsemantik erweitern wir zuerst die Substitutionsoperation:

$$(\texttt{rec}\ x_1(x_2 : t_2) : t_1\ \texttt{=>}\ e)\,[e'/x] = \begin{array}{l}\text{if } x = x_1 \vee x = x_2 \\ \text{then } \texttt{rec}\ x_1(x_2 : t_2) : t_1\ \texttt{=>}\ e \\ \text{else } \texttt{rec}\ x_1(x_2 : t_2) : t_1\ \texttt{=>}\ (e\,[e'/x])\end{array}$$

Zusätzlich fügen wir eine Auswertungsregel für rekursive Abstraktionen hinzu:

$$\frac{e' = (\texttt{fn}\ x_2 : t_2\ \texttt{=>}\ e)\,[\texttt{rec}\ x_1(x_2 : t_2) : t_1\ \texttt{=>}\ e\ /\ x_1]}{\texttt{rec}\ x_1(x_2 : t_2) : t_1\ \texttt{=>}\ e\ \rightarrow\ e'}$$

Rekursive Abstraktionen sind also keine kanonischen Ausdrücke.

Mit rekursiven Abstraktionen kann man zulässige Ausdrücke schreiben, deren Auswertung divergiert. Folglich sind der Auswertbarkeitsatz, der Terminierungssatz und der Normalisierbarkeitssatz nicht auf FR übertragbar. Alle anderen Sätze und Propositionen gelten jedoch auch für FR.

Bei der Erweiterung von F um rekursive Prozeduren haben wir sehr von der modularen Struktur der Definition von F profitiert: Die existierenden Teile konnten ohne Überarbeitung mit den neuen Teile kombiniert werden.

Übungen

Aufgabe 11.1

1. Geben Sie einen Ausdruck von F an, der weder reduzierbar noch kanonisch ist.
2. Stellen Sie die lexikalischen Bindungen der FR Ausdrücke

```
rec f (n:int) : int => if n<=1 then 1 else n * f (n-1)
```

```
rec x (x:int) : int->int => fn y:int => z x y
```

durch Überstreichen und Indizieren von Bezeichnerauftreten dar.

3. Beschreiben Sie die zwei obigen FR Ausdrücke durch zwei Standard ML Ausdrücke des Typs `exp`.
4. Geben Sie den Ausdruck an, den die folgende Substitution liefert:

   $((x\ 5) + (x\ 3))\ [\texttt{fn}\ x : \texttt{int} \Rightarrow x\ /\ x]$

Aufgabe 11.2 (Geschlossene Ausdrücke) Schreiben Sie eine Prozedur

```
closed : exp -> bool
```

die testet, ob ein Ausdruck geschlossen ist. Verwenden Sie eine Hilfsprozedur

```
closed' : exp -> id list -> bool
```

die testet, ob alle freien Bezeichner eines Ausdrucks in einer Liste vorkommen. Legen Sie die Ausdrücke von FR zugrunde.

Aufgabe 11.3 (Freie Bezeichner) Schreiben Sie eine Prozedur

```
freeIds : exp -> id list
```

die zu einem Ausdruck eine Liste der frei auftretenden Bezeichner liefert. Die Liste darf denselben Bezeichner mehrfach enthalten. Verwenden Sie eine Hilfsprozedur

```
freeIds' : id list -> exp -> id list
```

die nur die frei auftretenden Bezeichner liefert, die nicht in einer Liste von gebundenen Bezeichnern enthalten sind. Legen Sie die Ausdrücke von FR zugrunde.

Aufgabe 11.4 (Umgebungen) Schreiben Sie eine Strukturdeklaration, die Umgebungen gemäß der Signatur in Abbildung 11.6 implementiert. Implementieren Sie Umgebungen durch Prozeduren

```
type 'a env = id -> 'a
```

die die Ausnahme `Unbound` werfen, wenn sie auf Bezeichner angewendet werden, denen sie keinen Wert zuordnen.

Aufgabe 11.5 (F mit Rekursion) Sie sollen die Implementierung der statischen und der dynamischen Sematik von F um rekursive Abstraktionen erweitern. Geben Sie die entsprechenden Erweiterungen der Prozeduren `elab` und `eval` an.

Aufgabe 11.6 (Einschritt-Semantik) Sie sollen die Einschritt-Semantik für FR implementieren. Schreiben Sie dazu Prozeduren

```
canonical : exp -> bool
subst     : exp -> exp -> id -> exp
reduceOp  : con -> ops -> con -> exp  (* Error *)
reduce    : exp -> exp  (* Error *)
normalize : exp -> exp  (* Error *)
```

wie folgt:

1. `canonical` testet, ob ein Ausdruck kanonisch ist.
2. `subst` liefert zu e, e' und x den Ausdruck $e\,[e'/x]$.
3. `reduceOp` liefert zu c_1, o und c_2 einen kanonischen Ausdruck, der das Ergebnis der Operationsanwendung $c_1\ o\ c_2$ beschreibt.
4. `reduce` versucht, einen Ausdruck einmal mit $\rightarrow$ zu reduzieren. Wenn der Ausdruck nicht reduzierbar ist, wird die Ausnahme `Error` geworfen.
5. `normalize` versucht, die Normalform eines Ausdrucks zu berechnen. Wenn keine Normalform existiert, aber der Ausdruck nur endlich oft reduziert werden kann, wird die Ausnahme `Error` geworfen.

Aufgabe 11.7 (Erweiterung von F um Andalso) Wir wollen F um Ausdrücke der Form

e_1 `andalso` e_2

erweitern. Da sich solche Ausdrücke als abgeleitete Form für

`if` e_1 `then` e_2 `else false`

darstellen lassen, ist eigentlich keine Erweiterung der abstrakten Syntax erforderlich. Zu Übungszwecken sei die abstrakte Syntax aber wie folgt erweitert:

$e \in \mathit{Exp} = \cdots \mid e_1$ `andalso` e_2

1. Geben Sie die für die statische Semantik zusätzlich erforderlichen Inferenzregeln an.
2. Geben Sie die für die dynamische Semantik zusätzlich erforderlichen Inferenzregeln an.
3. Geben Sie die für die Einschritt-Semantik zusätzlich erforderlichen Inferenzregeln an.

Aufgabe 11.8 (Erweiterung von F um Paare) Wir wollen F um Paare erweitern. Die abstrakte Syntax und die Menge der Werte sollen dafür wie folgt erwei-

tert werden:

$$t \in Ty = \cdots \mid t_1 * t_2$$
$$e \in Exp = \cdots \mid (e_1, e_2) \mid \texttt{fst}\ e \mid \texttt{snd}\ e$$
$$v \in Val = \mathbb{Z} \cup Pro \cup Val \times Val$$

1. Geben Sie die für die statische Semantik zusätzlich erforderlichen Inferenzregeln an.
2. Geben Sie die für die dynamische Semantik zusätzlich erforderlichen Inferenzregeln an.
3. Definieren Sie die Menge der kanonischen Terme und geben Sie die für die Einschritt-Semantik zusätzlich erforderlichen Inferenzregeln an.

Aufgabe 11.9 (Erweiterung von F um Listen) Wir wollen F um Listen erweitern. Die abstrakte Syntax und die Menge der Werte sollen dafür wie folgt erweitert werden:

$$t \in Ty = \cdots \mid t\ list$$
$$e \in Exp = \cdots \mid \texttt{nil}\colon t \mid e_1 :: e_2 \mid \texttt{null}\ e \mid \texttt{hd}\ e \mid \texttt{tl}\ e$$
$$v \in Val = \mathbb{Z} \cup Pro \cup Val \times Val$$

Der Ausdruck `nil`: t beschreibt die leere Liste und gibt für diese den Typ t vor. Die leere Liste soll durch die Zahl 0 dargestellt werden.

1. Geben Sie die für die statische Semantik zusätzlich erforderlichen Inferenzregeln an.
2. Geben Sie die für die dynamische Semantik zusätzlich erforderlichen Inferenzregeln an.
3. Definieren Sie die Menge der kanonischen Terme und geben Sie die für die Einschritt-Semantik zusätzlich erforderlichen Inferenzregeln an.

Aufgabe 11.10 (Challenge: Fakultät ohne Rekursion) Geben Sie in F einen geschlossenen Ausdruck *fak* an, der zu einer Prozedur evaluiert, die die Fakultätsfunktion berechnet. Sie sollen dabei keine rekursive Abstraktion benutzen. Stattdessen soll *fak* Rekursion mithilfe von Selbstapplikation $x\ x$ simulieren. Daher wird es keinen Typ t mit $\emptyset \vdash fak \Rightarrow t$ geben. Beginnen Sie mit dem Ausdruck

```
fn f:int->int => fn n:int =>
  if n<=1 then 1 else n * f (n-1)
```

Überzeugen Sie sich davon, dass Ihr Ausdruck *fak* tut was er soll, indem Sie *fak n* für einige $n \in \mathbb{N}$ mit `eval` und `normalize` (siehe Aufgabe 11.6) auswerten.